

Calculation of the relative speed of sound in a gas mixture

MICHAEL J. ACKERMAN and GEORGIA MAITLAND

Behavioral Sciences Department, U.S. Naval Medical Research Institute, Bethesda, Maryland 20014

Ackerman, M. J., and G. Maitland, 1975. Calculation of the relative speed of sound in a gas mixture. *Undersea Biomed. Res.* 2(4):305-310.—Since the frequency spectrum of a voice signal is directly dependent on the velocity of sound, studies of speech spectra include the problem of calculating the speed of sound in the gas mixture being used. A computer program written in BASIC has been developed to calculate the speed of sound relative to air in various diving gas mixtures. In addition, a set of tables available as a separate technical report has been generated using this program. These tables are designed to provide a standard reference for reporting spectral shifts in speech due to different gas mixtures under normal diving conditions.

speed of sound
K factor

density
gas mixtures

Investigators studying speech spectra face the problem of calculating the speed of sound in the gas mixture with which they are working. The frequency spectrum of a voice signal is directly dependent on the velocity of sound. Comparisons of reported results from previous studies reveal discrepancies, which may emanate largely from errors in the calculation of the speed of sound in the gas mixtures. The present program has developed computer-generated tables designed to provide a standard reference for reporting spectral shifts in speech due to different gas mixtures, under normal diving conditions. Any study dealing with acoustic signals and different gas mixtures (Sergeant 1963; MacLean 1965; Maitland and Findling 1974; Maitland and Thomas 1974) would find such a table quite useful.

DERIVATION OF *K*

Fant and Lindqvist (1968) designated a factor, *K*, defined as the ratio of the velocity of sound in a gas mixture to that of air. It can be shown that this factor *K* depends only on the ratio of the specific heat ratios (γ) and the molecular weights for air and the gas.

The speed of sound in a gas of mass density ρ at a pressure *P* is given by

$$S = \left(\frac{\gamma P}{\rho} \right)^{1/2} \quad (1)$$

Symbols used to calculate K are defined in the following list:

SYMBOLS

A	subscript for air
ρ	density
G	subscript for gas mixture
$K = \frac{S_G}{S_A}$	K factor, speed of sound ratio
M	Molecular weight
P	pressure
P_i	percent volume present of gas i
R	gas constant
S	speed of sound
T	temperature ($^{\circ}\text{K}$)
$\gamma = \frac{C_P}{C_V}$	specific heat ratio
C_P	specific heat at constant pressure
C_V	specific heat at constant volume

Since the mass density can be written in terms of the molecular weight of the gas, the pressure, and the absolute temperature as

$$\rho = \frac{MP}{RT}, \quad (2)$$

the speed of sound in the gas is

$$S = \left(\frac{\gamma RT}{M} \right)^{1/2}. \quad (3)$$

Note that S is independent of the pressure. Thus, for air

$$S_A = \left(\frac{\gamma_A RT}{M_A} \right)^{1/2} \quad (4)$$

and for a gas

$$S_G = \left(\frac{\gamma_G RT}{M_G} \right)^{1/2} \quad (5)$$

By definition

$$K = S_G/S_A \quad (6)$$

so that

$$K = \left(\frac{\gamma_G M_A}{\gamma_A M_G} \right)^{1/2} \quad (7)$$

The K factor depends only on the ratio of the γ 's and molecular weights for air and the gas.

COMPUTER PROGRAM

Table 1 gives the values of γ , M , and C_V for various diving gases. Values for γ and M for mixed gases may be found by the following formulae (Handel and Strange 1966).

$$M_G = \sum p_i M_i \quad (8a)$$

$$\gamma_G = 1 + \frac{R}{\sum p_i M_i C_{Vi}} \quad (8b)$$

Inserting the values for M_A and γ_A into Equation 7 yields

$$K = \left(\frac{20.6752 \gamma_G}{M_G} \right)^{1/2} = 4.547 \left(\frac{\gamma_G}{M_G} \right)^{1/2} \quad (9)$$

Table 2 gives a BASIC program using Equation 9 for generating values of K for any mixture of helium, neon, oxygen, hydrogen, and nitrogen under normal diving conditions.

TABLE 1

Values for M , C_V , and γ for air and five diving gases*

Gas	M	$C_V^\dagger$	γ
O ₂	31.9988	0.1554	1.401
He	4.003	0.752	1.660
N ₂	28.0134	0.1765	1.404
Ne	20.183	0.15	1.64
H ₂	2.016	2.402	1.41
Air	28.966	0.1710	1.401

*After *CRC Handbook of Chemistry & Physics* (West 1973) and *Handbook of Chemistry* (Lange 1967).

†Specific heat at constant volume in cal/mg/°C.

TABLE 2

Basic program to calculate K .

```

0010 REM      ***BASIC PROGRAM TO CALCULATE "K" ***
0020 REM
0030 REM      O2 = PERCENT OXYGEN
0040 REM      N2 = PERCENT NITROGEN
0050 REM      H2 = PERCENT HYDROGEN
0060 REM      H  = PERCENT HELIUM
0070 REM      N  = PERCENT NEON
0080 REM      M  = MOLECULAR WEIGHT
0090 REM      K  = "K" FACTOR
0100 REM      R  = GAS CONSTANT IN CAL/DEG C
0110 REM      G  = SPECIFIC HEAT RATIO "GAMMA"
0120 REM
0130 LET R=1.9868
0140 REM
0150 REM      -----INPUT PERCENT OF EACH GAS-----
0160 REM
0170 PRINT
0180 PRINT "ENTER PERCENT OXYGEN (-1 TO END)";
0190 INPUT O2
0200 PRINT
0210 IF O2<0 THEN GOTO 0290
0220 PRINT "ENTER PERCENT NITROGEN";
0230 INPUT N2
0240 PRINT
0250 PRINT "ENTER PERCENT HELIUM";
0260 INPUT H
0270 PRINT
0280 PRINT "ENTER PERCENT NEON";
0290 INPUT N
0300 PRINT
0310 PRINT "ENTER PERCENT HYDROGEN";
0320 INPUT H2
0330 PRINT
0340 REM
0350 REM      ---CHECK THAT PERCENTAGES ADD TO 100%---

```

```

0360 REM
0370 IF N+H+N2+O2+H2=100 THEN GOTO 0440
0380 PRINT "DOES NOT ADD-UP TO 100%"
0390 PRINT
0400 GOTO 0170
0410 REM
0420 REM ----- CALCULATE 'K' -----
0430 REM
0440 LET O2=O2/100
0450 LET N2=N2/100
0460 LET H=H/100
0470 LET N=N/100
0480 LET H2=H2/100
0490 LET M=31.9988*O2+4.003*H+28.0134*N2+20.183*N+2.016*H2
0500 LET K1=4.97261*O2+3.01026*H+4.94437*N2+3.02745*N+4.84243*H2
0510 LET G=1+R/K1
0520 LET K=4.547*S0R(G/M)
0530 REM
0540 REM ----- PRINT CALCULATED VALUE OF 'K' -----
0550 REM
0560 PRINT
0570 PRINT 'K = 'JK
0580 GOTO 0170
0590 END
0600 REM *****

```

TABLE 3

Examples of calculation of K using the basic program.

RUN

```

ENTER PERCENT OXYGEN (-1 TO END) ? 20
ENTER PERCENT NITROGEN ? 80
ENTER PERCENT HELIUM ? 0
ENTER PERCENT NEON ? 0
ENTER PERCENT HYDROGEN ? 0

```

$K = 1.00283$

```

ENTER PERCENT OXYGEN (-1 TO END) ? 10
ENTER PERCENT NITROGEN ? 90
ENTER PERCENT HELIUM ? 5
ENTER PERCENT NEON ? 0
ENTER PERCENT HYDROGEN ? 0
DOES NOT ADD-UP TO 100%

```

```

ENTER PERCENT OXYGEN (-1 TO END) ? .5
ENTER PERCENT NITROGEN ? 5
ENTER PERCENT HELIUM ? 94.5
ENTER PERCENT NEON ? 0
ENTER PERCENT HYDROGEN ? 0

```

$K = 2.51703$

```

ENTER PERCENT OXYGEN (-1 TO END) ? .5
ENTER PERCENT NITROGEN ? 0
ENTER PERCENT HELIUM ? 99.5
ENTER PERCENT NEON ? 0
ENTER PERCENT HYDROGEN ? 0

```

$K = 2.87636$

```

ENTER PERCENT OXYGEN (-1 TO END) ? -1

```

END AT 0590

*OK

RESULTS

The derivation of the K factor has been provided. A BASIC program for generating values of K for any combination of oxygen, nitrogen, helium, hydrogen, and neon has been presented. Table 3 demonstrates the use of this program for calculating the K factor of some common gas mixtures. In addition, a series of tables giving values of K for various mixtures of the above gases (except hydrogen) are available as a technical report (Ackerman and Maitland 1974). It is hoped that this program and its associated tables will provide a standard reference for reporting spectral shifts in speech due to different gas mixtures.

From Bureau of Medicine and Surgery, Navy Department, Research Subtask M4306.03.3010DAC5. The opinions and statements contained herein are the private ones of the writer and are not to be construed as official or reflecting the views of the Navy Department or the naval service at large.

Received for publication May 1975; revised manuscript received August 1975.

Ackerman, M. J., and G. Maitland. 1975. Calcul de la vitesse relative du son dan un mélange gazeux. Undersea Biomed. Res. 2(4):305-310.—Puisque le spectre de la fréquence d'un signal vocal dépend directement de la vitesse du son, les études des spectres de la parole comprennent le problème du calcul de la vitesse du son dans le mélange gazeux employé. Un programme pour ordinateur en langue BASIC avait été développé pour calculer la vitesse du son relative à l'air dans des mélanges divers employés pour la plongée. De plus, des tables, publiées comme rapport technique séparé, ont été générées par ce programme. Ces tables forment une référence pour caractériser les altérations spectrales de la parole dues aux mélanges gazeux divers sous les conditions normales de la plongée.

vitesse du son
facteur K

densité
mélanges gazeux

REFERENCES

- Ackerman, M. J., and G. Maitland. 1974. A set of tables of the relative speed of sound in various diving gas mixtures. U.S. Naval Medical Research Institute, Bethesda, Maryland. AD780414.
- Fant, G., and J. Lindqvist. 1968. Pressure and gas mixture effects on diver's speech. Quarterly Progress & Statistics Report, Speech Translation Laboratory, Royal Institute of Technology (Stockholm) STL-QPSR-1/1968:7-20.
- Handel, R. A., and M. G. Strange. 1966. Acoustic experiment to determine the composition of an unknown planetary atmosphere. J. Acoust. Soc. Am. 40(4):896-902.
- Lange, N. A., ed. 1967. Handbook of chemistry. 10th ed. McGraw-Hill, New York.
- MacLean, D. J. 1965. Analysis of speech in a helium-oxygen mixture under pressure. J. Acoust. Soc. Am. 40:625-627.
- Maitland, G., and A. Findling. 1974. Structural analysis of hyperbaric speech under three gases. Aerosp. Med. 45(4):380-385.
- Maitland, G., and J. R. Thomas. 1974. Frequency spectrum analysis of speech changes during open sea habitat and hyperbaric chamber dives. Aerosp. Med. 45(8):855-859.
- Sergeant, R. L. 1963. Speech during respiration of a mixture of helium and oxygen. Aerosp. Med. 34:826-829.
- West, R. C., ed. 1973. CRC handbook of chemistry and physics. 54th ed. Chemical Rubber Publishing Co., Cleveland, Ohio.